

# 科学绘图软件 Gnuplot

by leeyee  
SouthEast University

## 1, 简介

Gnuplot 是由 Colin Kelly 和 Thomas Williams 于 1986 年开始开发的科学绘图工具，遵守 GPL 软件许可证。支持多种平台：UNIX, IBM OS/2, MS Windows, DOS, Macintosh, VMS, Atari 等等。它的设计初衷是为了方便科学工作者和学生绘制函数曲线和其他数据图。现在，除了原本的绘图功能外，Gnuplot 还可以用作网页脚本、作为 Octave 等软件的第三方绘图引擎等。

Gnuplot 支持 2 维和 3 维图形，绘制如点、直线、曲线、曲面、向量场等图形。绘制的图形可以指定多种输出方式：交互图形界面、打印机输出、输出到文件等等。Gnuplot 和其他众多科学计算软件一样，支持交互命令行模式（Interactive mode）和脚本（Batch mode）两种工作模式。用户可以在其命令提示符下键入命令实时观察输出，也可以将命令写成脚本用 `gnuplot filename` 运行。

下面，我们就通过一些简单介绍来接触这个轻便而强大的工具。


## 2, Gnuplot 基础

Gnuplot 是一个基于命令行的软件，在终端下键入 `gnuplot` 后回车即可启动 Gnuplot。可以看到一个命令提示符 `gnuplot>`。下面我们就可以键入具体的命令来绘制我们想要的图形了。从最简单的开始，获取帮助的方法是在命令提示符下键入 `help ***` 查询。

绘制二维图形的命令是 `plot`，如：


```
gnuplot> plot sin(x)
```

即可看到我们非常熟悉的正弦函数曲线：



绘制三维图形的命令是 `splot`，如：

```
gnuplot> splot sin(y)*exp(-x**2-y**2)
```



在交互模式下，Gnuplot 对于生成的三维立体图形，可以通过鼠标的移动，使图形旋转、侧翻等等。

Gnuplot 支持众多的数学函数，常用的有：

<code>abs</code>	<code>arg</code>	<code>erfc</code>	<code>exp</code>
<code>int</code>	<code>log</code>	<code>log10</code>	<code>norm</code>
<code>rand</code>	<code>real</code>	<code>sgn</code>	<code>sqrt</code>
<code>cos</code>	<code>acos</code>	<code>cosh</code>	<code>acosh</code>
<code>sin</code>	<code>asin</code>	<code>sinh</code>	<code>asinh</code>
<code>tan</code>	<code>atan</code>	<code>thanh</code>	<code>atanh</code>

具体函数的细节，可以在命令提示符下键入 `help functionname` 查询。

在 gnuplot 中，可以用上面给出的函数构成复杂的函数表达式，绘制它们的图形。

Gnuplot 下加减乘除分别用 `+`, `-`, `*`, `/` 表示，`**X` 表示  $X$  次幂。

例如，我们绘制  $(-2,2)$  上的  $x^3$  曲线：

```
gnuplot> set grid
```

```
gnuplot> plot [-2:2] x**3
```

即可。其中，`set grid` 设定了输出的图形的网格线，而 `[-2:2]` 确定了曲线的  $x$  区间。

### 3. 定制外观

Gnuplot 是一个可定制性非常高的绘图软件，可以通过改变各类参数得到非常专业、美观的图形。Gnuplot 下使用 `set` 命令定制各种参数，下面是一些常见的定制参数。

- 设置取值区间

`xrange` 参数定义函数的 `x` 轴区间，如 `set xrange [-pi:pi]`。类似地，`yrange` 可以确定 `y` 轴的区间。


- 设置标题

```
set title "Name of the graph"
set xlabel "Name of the X axis"
set ylabel "Name of the Y axis"
```

- 设置特殊点标记

常用在绘制以角度为变量的函数中，就 `x` 轴上的关键点用角度代替原来的数字。例如：  
`set xtics ("90" pi/2, "-90" -pi/2)` 即表示将 `x` 轴上的  $-\pi/2$  和  $\pi/2$  两点用 `-90` 和 `90` 来表示。  
 针对上述三点，举例如下：

```
gnuplot> set xrange [-pi:pi]
gnuplot> set title "My first custom graph"
gnuplot> set xlabel "Angle, in degrees"
gnuplot> set ylabel "sin(angle)"
gnuplot> set xtics ("90" pi/2, "90" -pi/2)
gnuplot> set grid
gnuplot> plot sin(x)
```



- 定义曲线说明

注意到曲线右上角的曲线说明，Gnuplot 允许我们自定义这些说明，包括位置、名称、以及加框。

```
set key top left          #将说明移到左上角
set key box               #给说明加框
```

- 设定曲线样式


有时候，我们需要在同一张图上绘制两条或多条不同的曲线，Gnuplot 很容易做到这

一点。默认情况下，它会用不同的颜色来显示各条曲线。我们可以用 **with** 字句来定义不同的曲线样式(`help plot with`)。

举例如下：

```
gnuplot> set xrange [-pi:pi]
gnuplot> set key top left
gnuplot> set key box
gnuplot> plot sin(x) title "sinusoid" with linespoints pointtype 5, \
      cos(x) t 'cosine' w boxes lt 4
```

其中的\是换行符号。运行后得到：



#### • 一页多图

前面我们都是将两条曲线绘制在一个页面上的。Gnuplot 同样允许在一个页面上绘制几条相邻的曲线：`multiplot`。类似于 MATLAB 中的 `subplot` 命令。将下列内容写入一个脚本，命名为 `mlplt.gpt`。在 gnuplot 的命令提示符下输入：`load 'mlplt.gpt'`





```
set xrange [-2*pi:2*pi]
set grid
set size 1,1                                #设定窗口大小
set origin 0,0
set multiplot                                #设定 multiplot 模式
set size 0.5,0.5
set origin 0,0.5                             #绘制左上角图形
plot sin(x)
set size 0.5,0.5
set origin 0,0                                #绘制左下角图形
plot sin(x)/x
set size 0.5,0.5
set origin 0.5,0.5                           #绘制右上角图形
```

```

plot cos(x)
set size 0.5,0.5
set origin 0.5,0
plot cos(x)/x
unset multiplot
reset

```

#绘制右下角图形



#### •保存设置

Gnuplot 允许我们保存现有的设置参数供以后使用。主要的参数设置都是有 `set` 执行的，因此，我们可以用

```
save set 'filename'
```

来保存我们已有的设置。

另外，Gnuplot 在每次运行的时候都会去读 `home` 目录下的 `.gnuplot` 文件，如果存在这个文件，那么 Gnuplot 就会直接运行里面的命令。所以我们可以把自己的设定参数写在 `~/.gnuplot` 里，每次运行时自动装载。

#### 4, 重定向输出

Gnuplot 默认的图形输出是桌面，即 X11。实际上，Gnuplot 可以支持许多不同设备终端的输出，包括打印机、postscript、png 图片等等。

将图形输出到 png 文件：

```

set terminal png
set output "output.png"
plot sin(x)

```

运行后，Gnuplot 不再弹出图形窗口，而是在运行 `gnuplot` 前所在的目录生成一个 `output.png` 文件，而图片内容正是你所绘制的  $\sin(x)$  曲线。


关于更多的输出形式，参见 `help set terminal`。

## 5, 其他应用

### •极坐标曲线

Gnuplot 允许用户绘制用极坐标方程描述的函数曲线。以心形线  $\rho = a(1 + \cos \theta)$  为例, 取  $a=3$ :


```
gnuplot> set polar                #设定极坐标模式
gnuplot> set trange [0:2*pi]      #极坐标下参量为 t
gnuplot> plot 3*(1+cos(t))
```



### •参数方程

Gnuplot 同样允许绘制由参数方程给定的函数曲线, 以  $x_1(t)=\tan(t), y_1(t)=t$  和  $x_2(t)=t, y_2(t)=\tan(t)$  为例:

```
gnuplot> set parametric           #设定参数方程模式
gnuplot> set xrange [-5:5]
gnuplot> set yrange [-5:5]
gnuplot> plot tan(t), t, t, tan(t)
```



#### 6, 寻求帮助

在 **gnuplot** 的项目站点, 有许多值得一看的文档和示例程序

<http://gnuplot.sourceforge.net/>

Google group 中有一个 **gnuplot** 用户邮件列表

[comp.graphics.apps.gnuplot](mailto:comp.graphics.apps.gnuplot)

关于 **Gnuplot** 的 IRC 频道

FreeNode **#gnuplot**